

**Веденников Николай Викторович,  
Кротков Павел Андреевич,  
Ульянцев Владимир Игоревич**

## ЗАДАЧА «ЕГЭ»

Этой статьей мы продолжаем цикл публикаций олимпиадных задач для школьников по информатике. Решение таких задач и изучение разборов поможет Вам повысить уровень практических навыков программирования и подготовиться к олимпиадам по информатике.

В этой статье рассматривается задача «ЕГЭ», которая предлагалась на восьмой индивидуальной олимпиаде цикла интернет-олимпиад для школьников сезона 2011–2012. Материалы интернет-олимпиад можно найти на сайте <http://neerc.ifmo.ru/school/io/>.

### УСЛОВИЕ ЗАДАЧИ

Школьник Игорь с детства любил компьютер. Он любил на нем играть, смотреть фильмы и делать много других приятных вещей. Иногда он даже программировал на нем, так как у него была мечта поступить в университет и стать программистом. Но Игорь был не очень усердным учеником, поэтому день сдачи единого государственного экзамена (ЕГЭ) по информатике наступил для него совершенно неожиданно. А ведь без хорошего балла за ЕГЭ его мечта так и останется мечтой.

На пробном ЕГЭ ему попалось задание, решения которого Игорь не знал. Задание требовало выписать семь подряд идущих

символов  $k$ -ой строки  $S_k$ , начиная с ее  $i$ -го символа. Данные строки строились по следующим правилам:

$$\begin{aligned}S_0 &= 0; \\S_1 &= 01; \\S_2 &= 0110; \\S_3 &= 01101001;\end{aligned}$$

...

$$S_j = S_{j-1} + \text{inv}(S_{j-1}),$$

где  $\text{inv}(S)$  обозначает строку  $S$ , в которой все нули заменены на единицы и наоборот.

Иgorь просит Вас помочь решить эту задачу, написав программу для ее решения.

#### Формат входного файла

Во входном файле содержатся два целых числа –  $k$  и  $i$  ( $3 \leq k \leq 63$ ,  $1 \leq i \leq 2^k - 7$ ).



Формат выходного файла

В выходной файл требуется вывести семь подряд идущих символов  $k$ -ой строки, начиная с  $i$ -го символа. Нумерация символов в строке начинается с единицы.

Примеры входных и выходных данных

ege.in	ege.out
3 2	1101001
4 6	0011001

**Листинг 1.** Исходный код решения задачи

```
const
    MAXK = 63;
var
    i, k: longint;
    n: int64;
    len: array [1..MAXK] of int64; // len[i] - длина  $i$ -ой строчки
    ans: string;

// Функция solve находит  $i$ -й символ в  $k$ -й строке
function solve(k: longint; i: int64): char;
var
    inv: boolean;
begin
    inv := false; // inv - инвертируем символ или нет
    while (k > 0) do begin
        // Сведение задачи к меньшей размерности
        if (i > len[k]) then begin
            i := i - len[k];
            inv := not inv;
        end;
        k := k - 1;
    end;
    if (inv) then
        solve := '1'
    else
        solve := '0';
end;

begin
    assign(input, 'ege.in');
    reset(input);
    assign(output, 'ege.out');
    rewrite(output);

    read(k, n);
    len[1] := 1;
    for i := 2 to k do
        len[i] := 2 * len[i - 1];
    for i := 0 to 6 do
        ans := ans + solve(k, n + i);
    writeln(ans);
end.
```

## РАЗБОР ЗАДАЧИ

Заметим, что, так как длина каждой строки в два раза больше длины предыдущей, длина  $k$ -ой строки равна  $2^k$ . При данных в задаче ограничениях длина строки  $|S_k| = 9\ 223\ 372\ 036\ 854\ 775\ 808$  при  $k = 63$ , поэтому решение задачи, основанное на непосредственном построении соответствующей строки, не будет удовлетворять ограничениям на время работы и затраты памяти программы.

Верное решение задачи основано на последовательном сведении задачи к задаче меньшей размерности. Решим задачу нахождения  $i$ -го символа в  $k$ -ой строке. Из условия задачи следует, что если  $k = 0$ , а  $i = 1$ , то этот символ – «0».

Теперь рассмотрим случай  $k > 1$ . Заметим, что если  $i < |S_{k-1}|$ , то искомый символ

содержится в строке  $S_{k-1}$ , и нам достаточно найти  $i$ -й символ в строке  $S_{k-1}$ . В противном случае искомый символ содержится в строке  $\text{inv}(S_{k-1})$ , и нам требуется найти в ней символ с номером  $(i - |S_{k-1}|)$ , для чего найдем данный символ в строке  $S_{k-1}$  и заменим его на противоположный. Необходимость замены символа будем хранить в переменной  $\text{inv}$ .

Таким образом, мы всякий раз сводим задачу к меньшей размерности, уменьшая  $k$  на единицу. Из этого следует, что время обработки каждой цифры составляет  $O(k)$ . Вычислив последовательно семь цифр искомой строки, получим требуемую в условии строку.

В листинге 1 приведена программная реализация решения задачи на языке программирования *Pascal*.

*Ведерников Николай Викторович,  
студент третьего курса кафедры  
«Компьютерные технологии»  
НИУ ИТМО, член жюри Интернет-  
олимпиад по информатике*

*Кротков Павел Андреевич,  
студент третьего курса кафедры  
«Компьютерные технологии»  
НИУ ИТМО, член жюри Интернет-  
олимпиад по информатике,*

*Ульянцев Владимир Игоревич,  
студент шестого курса кафедры  
«Компьютерные технологии»  
НИУ ИТМО, член жюри Интернет-  
олимпиад по информатике.*



Наши авторы, 2012.  
Our authors, 2012.